Politecnico di Milano
Formal Methods for Concurrent and Real-Time Systems

Computer Controller Automatic Transmission
(Mandatory Part, Optional Part coming soon)

Alessandra Bonetto
Filippo Sironi
Matteo Villa

2009

Contents

List of Figures

Listings

1 CCAT Class

The *ComputerControlledAutomaticTransmission* class is formalized thanks to the code reported in Listing 1 while Figure 1 shows the *big picture* of our complete designed.

Listing 1: ComputerControlledAutomaticTransmission.trio

```
class ComputerControlledAutomaticTransmission

import:
    HydraulicSystem,
    PlanetaryGearSet,
    TransmissionControlUnit,
    VehicleSpeedSensor,
    EngineSpeedSensor;

signature:

visible:
    torqueConverterState,
    vehicleSpeed,
    engineSpeed;

temporal domain: real;

domains:
    TorqueConverterState: {Attached, Detached};

items:
    TD total torqueConverterState: TorqueConverterState;
    TD total vehicleSpeed: integer;
    TD total engineSpeed: integer;

modules:
    hydraulicSystem: HydraulicSystem;
    planetaryGearSet: PlanetaryGearSet;
    transmissionControlUnit: TransmissionControlUnit;
    vehicleSpeedSensor: VehicleSpeedSensor;
    engineSpeedSensor: EngineSpeedSensor;

connections:
    (direct EngineSpeedSensor.actualSpeed, engineSpeed)
    (direct vehicleSpeedSensor.actualSpeed, vehicleSpeed)
```

```
37      (direct planetaryGearSet.transmissionShaftState,
            torqueConverterState)
38
39      (direct planetaryGearSet.gearShift,
            hydraulicSystem.gearShift)
40      (direct planetaryGearSet.gearDrive,
            hydraulicSystem.gearDrive)
41      (direct planetaryGearSet.gearPark,
            hydraulicSystem.gearPark)
42      (direct planetaryGearSet.gearReverse,
            hydraulicSystem.gearReverse)
43      (direct hydraulicSystem.controlGearShift,
            transmissionControlUnit.controlGearShift)
44      (direct transmissionControlUnit.receiveEngineSpeed,
            engineSpeedSensor.sendSpeed)
45      (direct transmissionControlUnit.receiveVehicleSpeed,
            vehicleSpeedSensor.sendSpeed)
46
47  end
```



Figure 1: Computer Controller Automatic Transmission

Due to limitations in the verification tools we have decided to focus our verification process only on the Hydraulic System, described in Section 4, and on the Planetary Gear Set, described in Section 3.

The specification is reported in Listing 2.

Listing 2: ComputerControlledAutomaticTransmission.t2p

```
variables
    controlGearShift:[0..4],
    gearHandle:[0..3],
    gearShift:[0..7],
    actualGear:[1..5],
    transmissionShaftState:[0..1]

constants
    noEvent=0,
    TCUOneUp = 1, TCUTwoUp=2, TCUOneDown=3, TCUTwoDown=4,
    handleDrive=1, handlePark=2, handleReverse=3,
    shiftDrive=1, shiftOneUp=2, shiftTwoUp=3,
       shiftOneDown=4, shiftTwoDown=5, shiftPark=6,
       shiftReverse=7,
    first=0, second=1, third=2, park=3, reverse=4,
    detached=0, attached=1,
    fluidPropagationDelay=1, singleGearShiftDelay=2,
       dualGearShiftDelay=3, driveGearShiftDelay=3,
       parkGearShiftDelay=3, reverseGearShiftDelay=3

axioms
    Start:
        Now(gearHandle=handleDrive & actualGear=first);

    Mechanics:
        gearShift=shiftOneUp | gearShift=shiftTwoUp ->
           transmissionShaftState=attached;

    PropagateGearShiftCommand:
        (controlGearShift=TCUOneUp |
           controlGearShift=TCUTwoUp |
           controlGearShift=TCUOneDown |
           controlGearShift=TCUTwoDown ->
           Lasts_ii(gearHandle=noEvent,
           fluidPropagationDelay) &
           Lasts_ei(controlGearShift=noEvent,
           fluidPropagationDelay)) &
```

```
26          (controlGearShift=TCUOneUp ->
                Futr(gearShift=shiftOneUp,
                fluidPropagationDelay)) &
27          (controlGearShift=TCUTwoUp ->
                Futr(gearShift=shiftTwoUp,
                fluidPropagationDelay)) &
28          (controlGearShift=TCUOneDown ->
                Futr(gearShift=shiftOneDown,
                fluidPropagationDelay)) &
29          (controlGearShift=TCUTwoDown ->
                Futr(gearShift=shiftTwoDown,
                fluidPropagationDelay));
30
31      GearHandleCommand:
32          (gearHandle=handleDrive ->
                Lasts_ii(controlGearShift=noEvent,
                fluidPropagationDelay) &
                Lasts_ei(gearHandle=noEvent,
                fluidPropagationDelay) &
                Futr(gearShift=shiftDrive,
                fluidPropagationDelay)) &
33          (gearHandle=handlePark ->
                Lasts_ii(controlGearShift=noEvent,
                fluidPropagationDelay) &
                Lasts_ei(gearHandle=noEvent,
                fluidPropagationDelay) &
                Futr(gearShift=shiftPark,
                fluidPropagationDelay)) &
34          (gearHandle=handleReverse ->
                Lasts_ii(controlGearShift=noEvent,
                fluidPropagationDelay) &
                Lasts_ei(gearHandle=noEvent,
                fluidPropagationDelay) &
                Futr(gearShift=shiftReverse,
                fluidPropagationDelay));
35
36      GearShiftDrive:
37          gearShift=shiftDrive ->
                transmissionShaftState=detached;
38
39      GearShiftFirst:
40          (actualGear=first -> gearShift=noEvent |
                gearShift=shiftOneUp | gearShift=shiftTwoUp |
```

```
            gearShift=shiftPark | gearShift=shiftReverse) &
41          (actualGear=first & gearShift=shiftOneUp ->
               Lasts_ei(gearShift=noEvent,
               singleGearShiftDelay) & Futr(actualGear=second,
               singleGearShiftDelay)) &
42          (actualGear=first & gearShift=shiftTwoUp ->
               Lasts_ei(gearShift=noEvent, dualGearShiftDelay)
               & Futr(actualGear=third, dualGearShiftDelay)) &
43          (actualGear=first & gearShift=shiftPark ->
               Lasts_ei(gearShift=noEvent, parkGearShiftDelay)
               & Futr(actualGear=park, parkGearShiftDelay)) &
44          (actualGear=first & gearShift=shiftReverse ->
               Lasts_ei(gearShift=noEvent,
               reverseGearShiftDelay) & Futr(actualGear=third,
               reverseGearShiftDelay));
45
46      GearShiftSecond:
47          (actualGear=second -> gearShift=noEvent |
               gearShift=shiftOneUp | gearShift=shiftOneDown) &
48          (actualGear=second & gearShift=shiftOneUp ->
               Lasts_ei(gearShift=noEvent,
               singleGearShiftDelay) & Futr(actualGear=third,
               singleGearShiftDelay)) &
49          (actualGear=second & gearShift=shiftOneDown ->
               Lasts_ei(gearShift=noEvent,
               singleGearShiftDelay) & Futr(actualGear=first,
               singleGearShiftDelay));
50
51      GearShiftThird:
52          (actualGear=third -> gearShift=noEvent |
               gearShift=shiftOneDown |
               gearShift=shiftTwoDown) &
53          (actualGear=third & gearShift=shiftOneDown ->
               Lasts_ei(gearShift=noEvent,
               singleGearShiftDelay) & Futr(actualGear=second,
               singleGearShiftDelay)) &
54          (actualGear=third & gearShift=shiftTwoDown ->
               Lasts_ei(gearShift=noEvent, dualGearShiftDelay)
               & Futr(actualGear=first, dualGearShiftDelay));
55
56      GearShiftReverse:
57          (actualGear=reverse -> gearShift=noEvent |
               gearShift=shiftDrive | gearShift=shiftPark) &
```

```
58          (actualGear=reverse & gearShift=shiftDrive ->
               Lasts_ei(gearShift=noEvent,
               driveGearShiftDelay) & Futr(actualGear=first,
               driveGearShiftDelay)) &
59          (actualGear=reverse & gearShift=shiftPark ->
               Lasts_ei(gearShift=noEvent, parkGearShiftDelay)
               & Futr(actualGear=park, parkGearShiftDelay)) &
60          (gearShift=shiftReverse ->
               transmissionShaftState=detached);
61
62      GearShiftPark:
63          (actualGear=park ->
               transmissionShaftState=detached) &
64          (actualGear=park -> gearShift=noEvent |
               gearShift=shiftDrive | gearShift=shiftReverse) &
65          (actualGear=park & gearShift=shiftDrive ->
               Lasts_ei(gearShift=noEvent,
               driveGearShiftDelay) & Futr(actualGear=first,
               driveGearShiftDelay)) &
66          (actualGear=park & gearShift=shiftReverse ->
               Lasts_ei(gearShift=noEvent,
               reverseGearShiftDelay) &
               Futr(actualGear=reverse,
               reverseGearShiftDelay)) &
67          (gearShift=shiftReverse ->
               transmissionShaftState=detached);
```

2 Vehicle/EngineSpeedSensor Classes

The *VehicleSpeedSensor* is formalized thanks to the code reported in Listing 3 while the *EngineSpeedSensor* is formalized thanks to the code reported in Listing 4.

During the formalization of sensors we decided to simplify the design assuming that every time a `sampleSpeed` event occurs the state variable `actualSpeed` - which is `time dependent` and `total` - is automatically updated with the actual measured speed. This means we don't provide any axioms formalizing this behavior.

Moreover, we specified the starting point of the constant frequency sample chain saying that sometimes in the past there was a `sampleSpeed` occurrence. Further more, we guarante that `sampleSpeed` events will accure at constant frequency. In addition, if the sensor has memory we imposed that the `storedValue` is equal to 0. These can be consider just like the "initial conditions" of the system.

At the end, we guaranteed a sensor performs the needed action if and only if a sample event occur.

We didn't write any axioms specifying the fact that a `sendSpeed` event is mutually exclusive with itself due to the `total time dependent` parameter it accepts.

Listing 3: VehicleSpeedSensor.trio

```
1  class VehicleSpeedSensor (const sampleInterval, const
       sampleDelay)
2
3  signature:
4
5  visible:
6      actualSpeed,
7      sendSpeed;
8
9  temporal domain: real;
10
11 items:
12     TI sampleInterval: real;
13     TI sampleDelay: real;
14     TD total storedSpeed: integer;
15     TD total actualSpeed: integer;
16     event sendSpeed (integer, integer);
17     event sampleSpeed;
18
```

```
19 axioms:
20 vars:
21     deltaSpeed: integer;
22     speed: integer;
23 formulae:
24     SpeedValues:
25         actualSpeed >= 0 and storedSpeed >= 0;
26
27     BeginSample:
28         SomP (storedSpeed = 0 & sampleSpeed);
29
30     SamplingDefinition:
31         sampleSpeed implies Futr (sampleSpeed,
               sampleInterval) and not Lasts (sampleSpeed,
               sampleInterval);
32
33     SamplingAction:
34         sampleSpeed implies Futr (deltaSpeed = actualSpeed
               - storedSpeed and speed = actualSpeed and
               sendSpeed (deltaSpeed, speed) and Lasts
               (storedSpeed = actualSpeed, sampleInterval),
               sampleDelay);
35
36     SendSpeed:
37         deltaSpeed = actualSpeed - storedSpeed and
               actualSpeed = speed and sendSpeed (deltaSpeed,
               speed) implies Past (sampleSpeed, sampleDelay);
38
39 end
```

Listing 4: EngineSpeedSensor.trio

```
1  class EngineSpeedSensor (const sampleInterval, const
       sampleDelay)
2
3  signature:
4
5  visible: actualSpeed, sendSpeed;
6
7  temporal domain: real;
8
9  items:
10     TI sampleInterval: real;
11     TI sampleDelay: real;
```

```
12      TD total actualSpeed: integer;
13      event sendSpeed (integer);
14      event sampleSpeed;
15
16  axioms:
17  vars:
18      speed: integer;
19  formulae:
20      SpeedValues:
21          actualSpeed >= 0;
22
23      BeginSample:
24          SomP (sampleSpeed);
25
26      SamplingDefinition:
27          sampleSpeed implies Futr (sampleSpeed,
                sampleInterval) and not Lasts (sampleSpeed,
                sampleInterval);
28
29      SampleSpeedActions:
30          sampleSpeed implies Futr (actualSpeed = speed and
                sendSpeed (speed), sampleDelay);
31
32      SendSpeed:
33          actualSpeed = speed and sendSpeed (speed) implies
                Past (sampleSpeed, sampleDelay);
34
35  end
```

3 PlanetaryGearSet Class

The *PlanetaryGearSet* class is formalized thanks to the code reported in Listing 5.

The Planetary Gear Set guarantees that every time a gear shift event occurs the `actualGear` will be maintained until the shift is finished.

Inside this component are defined all axioms limiting gear shifts to effective ones only (e.g. it is impossibile to shift down a gear if `actualGear` is `First`). The Planetary Gear Set permits to shift up to two gear at the same time (as the specification asks), however, the Transmission Control Unit doesn't use this possibility because in a real Planetary Gear Set this is not possibile.

Moreover, through the formalization of the Planetary Gear Set we impose that we can't receive a gear shift event if we are in the middle of a gear shift. Different gear shifting times are defined for different gears and different steps.

The gears `Drive`, `Park`, and `Reverse` can be selected if and only if the transmission shaft is decoupled from the engine.

The state of the Planetary Gear Set changes if and only if an event occurs.

Listing 5: PlanetaryGearSet.trio

```
1  class PlanetaryGearSet (const singleGearShiftDelay, const
       dualGearShiftDelay, const driveGearShiftDelay, const
       parkGearShiftDelay, const reverseGearShiftDelay)
2
3  signature:
4
5  visible:
6      actualGear,
7      transmissionShaftState,
8      gearShift,
9      gearDrive,
10     gearPark,
11     gearReverse,
12
13 temporal domain: real;
14
15 domains:
16     Gear: {First, Second, Third, Park, Reverse};
17     TransmissionShaftState: {Attached, Detached};
18     ShiftWidth: 1..2;
19     ShiftDirection: {Up, Down};
20
21 items:
22     TI singleGearShiftDelay: real;
```

```
23      TI dualGearShiftDelay: real;
24      TI driveGearShiftDelay: real;
25      TI parkGearShiftDelay: real;
26      TI reverseGearShiftDelay: real;
27      TD total actualGear: Gear;
28      TD total transmissionShaftState:
            TransmissionShaftState;
29      event gearShift (ShiftWidth, ShiftDirection);
30      event gearDrive;
31      event gearPark;
32      event gearReverse;
33
34  axioms:
35  vars:
36      gearShiftWidth: ShiftWidth;
37      gearShiftWidth2: ShiftWidth;
38      gearShiftDirection: ShiftDirection;
39      gearShiftDirection2: ShiftDirection;
40      gear: Gear;
41  formulae:
42      Mechanics:
43          all gearShiftWidth (gearShiftDirection = Up ->
                transmissionShaftState=Attached);
44
45      GearShiftDrive:
46          gearDrive implies transmissionShaftState =
                Detached;
47
48      GearShiftFirst:
49          (actualGear = First implies not gearDrive and not
                ex gearShiftWidth (gearShiftDirection = Down
                and gearShift (gearShiftWidth,
                gearShiftDirection))) and
50          (actualGear = First implies SomF (gearPark or
                gearReverse or ex gearShiftWidth,
                gearShiftDirection (gearShift (gearShiftWidth,
                gearShiftDirection)))) and
51          (actualGear = First and gearShiftWidth = 1 and
                gearShiftDirection = Up and gearShift
                (gearShiftWidth, gearShiftDirection) implies
                Lasts (actualGear = First,
                singleGearShiftDelay) and Futr (actualGear =
                Second, singleGearShiftDelay)) and
```

```
52          (actualGear = First and gearShiftWidth = 2 and
                gearShiftDirection = Up and gearShift
                (gearShiftWidth, gearShiftDirection) implies
                Lasts (actualGear = First, dualGearShiftDelay)
                and Futr (actualGear = Third,
                dualGearShiftDelay) and
53          (actualGear = First and gearPark implies Lasts
                (actualGear = First, parkGearShiftDelay) and
                Futr (actualGear = Park, parkGearShiftDelay))
                and
54          (actualGear = First and gearReverse implies Lasts
                (actualGear = First, reverseGearShiftDelay) and
                Futr (actualGear = Reverse,
                reverseGearShiftDelay));
55
56      GearShiftSecond:
57          (actualGear = Second implies not gearDrive and not
                gearPark and not gearReverse and not ex
                gearShiftDirection (gearShiftWidth = 2 and
                gearShift (gearShiftWidth,
                gearShiftDirection))) and
58          (actualGear = Second implies SomF (ex
                gearShiftDirection (gearShiftWidth = 1 and
                gearShift (gearShiftWidth,
                gearShiftDirection)))) and
59          (actualGear = Second and gearShiftWidth = 1 and
                gearShiftDirection = Up and gearShift
                (gearShiftWidth, gearShiftDirection) implies
                Lasts (actualGear = Second,
                singleGearShiftDelay) and Futr (actualGear =
                Third, singleGearShiftDelay)) and
60          (actualGear = Second and gearShiftWidth = 1 and
                gearShiftDirection = Down and gearShift
                (gearShiftWidth, gearShiftDirection) implies
                Lasts (actualGear = Second,
                singleGearShiftDelay) and Futr (actualGear =
                First, singleGearShiftDelay));
61
62      GearShiftThird:
63          (actualGear = Third implies not gearDrive and not
                gearPark and not gearReverse and not ex
                gearShiftWidth (gearShiftDirection = Up and
                gearShift (gearShiftWidth,
```

```
            gearShiftDirection))) and
64          (actualGear = Third implies SomF (ex
               gearShiftWidth (gearShiftDirection = Down and
               gearShift (gearShiftWidth,
               gearShiftDirection)))) and
65          (actualGear = Third and gearShiftWidth = 1 and
               gearShiftDirection = Down and gearShift
               (gearShiftWidth, gearShiftDirection) implies
               Lasts (actualGear = Third,
               singleGearShiftDelay) and Futr (actualGear =
               Second, singleGearShiftDelay)) and
66          (actualGear = Third and gearShiftWidth = 2 and
               gearShiftDirection = Down and gearShift
               (gearShiftWidth, gearShiftDirection) implies
               Lasts (actualGear = Third, dualGearShiftDelay)
               and Futr (actualGear = First,
               dualGearShiftDelay));
67
68      GearShiftReverse:
69          (actualGear = Reverse implies not gearReverse and
               all gearShiftWidth, gearShiftDirection (not
               gearShift (gearShiftWidth,
               gearShiftDirection))) and
70          (actualGear = Reverse implies SomF (gearDrive or
               gearPark)) and
71          (actualGear = Reverse and gearDrive implies Lasts
               (actualGear = Reverse, driveGearShiftDelay) and
               Futr (actualGear = First, driveGearShiftDelay))
               and
72          (actualGear = Reverse and gearPark implies Lasts
               (actualGear = Reverse, parkGearShiftDelay) and
               Futr (actualGear = Park, parkGearShiftDelay))
               and
73          (gearReverse implies transmissionShaftState =
               Detached);
74
75      GearShiftPark:
76          (actualGear = Park implies not gearPark and all
               gearShiftWidth, gearShiftDirection (not
               gearShift (gearShiftWidth,
               gearShiftDirection))) and
77          (actualGear = Park implies SomF (gearDrive or
               gearReverse)) and
```

```
78          (actualGear = Park and gearDrive implies Lasts
               (actualGear = Park, reverseGearShiftDelay) and
               Futr (actualGear = First, driveGearShiftDelay))
               and
79          (actualGear = Park and gearReverse implies Lasts
               (actualGear = Park, reverseGearShiftDelay) and
               Futr (actualGear = Reverse,
               reverseGearShiftDelay)) and
80          (actualGear = Park implies transmissionShaftState
               = Detached) and
81          (gearPark implies transmissionShaftState =
               Detached);
82
83      GearShiftTimings:
84          all gearShiftDirection ((actualGear = First or
               actualGear = Second or actualGear = Third) and
               gearShiftWidth = 1 and gearShift
               (gearShiftWidth, gearShiftDirection) implies
               not Lasts (gearDrive or gearPark or gearReverse
               or ex gearShiftWidth2, gearShiftDirection2
               (gearShift (gearShiftWidth2,
               gearShiftDirection2)), singleGearShiftDelay))
               and
85          all gearShiftDirection ((actualGear = First or
               actualGear = Third) and gearShiftWidth = 2 and
               gearShift (gearShiftWidth, gearShiftDirection)
               implies not Lasts (gearDrive or gearPark or
               gearReverse or ex gearShiftWidth2,
               gearShiftDirection2 (gearShift
               (gearShiftWidth2, gearShiftDirection2)),
               dualGearShiftDelay)) and
86          ((actualGear = Reverse and gearDrive) implies not
               Lasts (gearDrive or gearPark or gearReverse or
               ex gearShiftWidth2, gearShiftDirection2
               (gearShift (gearShiftWidth2,
               gearShiftDirection2)), driveGearShiftDelay)) and
87          ((actualGear = Reverse and gearPark) implies not
               Lasts (gearDrive or gearPark or gearReverse or
               ex gearShiftWidth2, gearShiftDirection2
               (gearShift (gearShiftWidth2,
               gearShiftDirection2)), parkGearShiftDelay)) and
88          ((actualGear = Park and gearDrive) implies not
               Lasts (gearDriver or gearPark or gearReverse or
```

```
            ex gearShiftWidth2, gearShiftDirection2
            (gearShift (gearShiftWidth2,
            gearShiftDirection2)), driveGearShiftDelay)) and
89          ((actualGear = Park and gearReverse) implies not
            Lasts (gearDrive or gearPark or gearReverse or
            ex gearShiftWidth2, gearShiftDirection2
            (gearShift (gearShiftWidth2,
            gearShiftDirection2)), reverseGearShiftDelay));
90
91      Nothing:
92          all gear (actualGear = gear and not (all
            gearShiftWidth, gearShiftDirection (gearShift
            (gearShiftWidth, gearShiftDirection)) or
            gearDrive or gearPark or gearReverse) implies
            UpToNow (actualGear = gear) and NowOn
            (actualGear = gear));
93
94  end
```

4 HydraulicSystem Class

The *HydraulicSystem* class is formalized thanks to the code reported in Listing 6.

The first assumption we made before modelling the Hydraulic System was that every valve and electrovalve configuration imposes the same fluid propagation delay; this means that for every command that the Hydraulic System propagates the delay will always be the same. This behavior is formalized with the `time independent constant fluidPropagationDelay`.

The *manual valve*, which permit the driver to manually select the gear mode, is modelled thanks to the `gearHandle` event and the `GearHandle` axiom. During the time in which the Hydraulic System propagate a command there can be no `gearHandle` event which somehow means the fluid propagation is faster then the driver reaction time (which is a realistic assumption).

Moreover, thanks to the `MutualExclusion` axiom, it's impossibile to generate two `gearHandle` event at the same time which means that the gear handle can't be for example in Park and Drive mode at the same instant.

Listing 6: HydraulicSystem.trio

```
1  class HydraulicSystem (const fluidPropagationDelay)
2
3  signature:
4
5  visible:
6      gearHandle,
7      gearShift,
8      gearDrive,
9      gearPark,
10     gearReverse,
11     controlGearShift;
12
13 temporal domain: real;
14
15 domains:
16     GearPosition: {Drive, Park, Reverse};
17     ShiftWidth: 1..2;
18     ShiftDirection: {Up, Down};
19
20 items:
21     TI fluidPropagationDelay: real;
22     event gearHandle (GearPosition);
23     event gearShift (ShiftWidth, ShiftDirection);
24     event gearDrive;
```

```
25      event gearPark;
26      event gearReverse;
27      event controlGearShift (ShiftWidth, ShiftDirection);
28
29  axioms:
30  vars:
31      gear: GearPosition;
32      gear2: GearPosition;
33      gearShiftWidth: ShiftWidth;
34      gearShiftWidth2: ShiftWidth;
35      gearShiftDirection: ShiftDirection;
36      gearShiftDirection2: ShiftDirection;
37  formulae:
38      GearHandleCommand:
39          (gear = Drive and gearHandle (gear) implies not
              Lasts (all gear2 (gear2 <> gear implies
              gearHandle (gear2) or ex gearShiftWidth,
              gearShiftDirection (controlGearShift
              (gearShiftWidth, gearShiftDirection))),
              fluidPropagationDelay) and Futr (gearDrive,
              fluidPropagationDelay) and
40          (gear = Park and gearHandle (gear) implies not
              Lasts (all gear2 (gear2 <> gear implies
              gearHandle (gear2) or ex gearShiftWidth,
              gearShiftDirection (controlGearShift
              (gearShiftWidth, gearShiftDirection))),
              fluidPropagationDelay) and Futr (gearPark,
              fluidPropagationDelay) and
41          (gear = Reverse and gearHandle (gear) implies not
              Lasts (all gear2 (gear2 <> gear implies
              gearHandle (gear2) or ex gearShiftWidth,
              gearShiftDirection (controlGearShift
              (gearShiftWidth, gearShiftDirection))),
              fluidPropagationDelay) and Futr (gearReverse,
              fluidPropagationDelay);
42
43      PropagateGearShiftCommand:
44          all gearShiftWidth, gearShiftDirection
              (controlGearShift (gearShiftWidth,
              gearShiftDirection) implies not Lasts
              (gearDrive or gearPark or gearReverse or ex
              gearShiftWidth2, gearShiftDirection2
              (controlGearShift (gearShiftWidth2,
```

```
            gearShiftDirection2)), fluidPropagationDelay)
            and Futr (gearShift (gearShiftWidth,
            gearShiftDirection), fluidPropagationDelay));
45
46      MutualExclusions:
47          all gear (gearHandle (gear) implies all gear2
              (gear <> gear2 implies not gearHandle (gear2)));
48
49  end
```

5 TransmissionControlUnit Class

The *TransmissionControlUnit* class is formalized thanks to the code reported in Listing 7.

Our first formalization of the Transmission Control Unit didn't take in account the possibility to have asynchronous sensors; the latest version of the Transmission Control Unit permits to manage asynchronous sensors thanks to internal memory modelled with three `time dependent total` values.

When handle the necessity to scale gears till the `First` with the assumption that the human reaction is way slower than sampling frequency and mechanical reactions, so, when the vehicle stops, the axiom which handle the gear scale manage to be "active" the necessary amount of times to scale all the gears.

The Transmission Control Unit guarantees that it doesn't raise more than one gear shift event per instant and it receives at most one event per instant from each sensor (this is described also in Section 2 and so guaranteed in VehicleSpeedSensor and EngineSpeedSensor class).

Listing 7: TransmissionControlUnit.trio

```
class TransmissionControlUnit

signature:

visible:
    controlGearShift,
    receiveEngineSpeed,
    receiveVehicleSpeed;

temporal domain: real;

domains:
    ShiftWidth: 1..2;
    ShiftDirection: {Up, Down};

items:
    TD total storedEngineSpeed: integer;
    TD total storedDeltaVehicleSpeed: integer;
    TD total storedVehicleSpeed: integer;
    event controlGearShift (ShiftWidth, ShiftDirection);
    event receiveEngineSpeed (integer);
    event receiveVehicleSpeed (integer, integer);

axioms:
vars:
```

```
26      engineSpeed: integer;
27      engineSpeed1: integer;
28      engineSpeed2: integer;
29      deltaVehicleSpeed: integer;
30      deltaVehicleSpeed1: integer;
31      deltaVehicleSpeed2: integer;
32      vehicleSpeed: integer;
33      vehicleSpeed1: integer;
34      vehicleSpeed2: integer;
35      gearShiftWidth1: ShiftWidth;
36      gearShiftWidth2: ShiftWidth;
37      gearShiftDirection1: ShiftDirection;
38      gearShiftDirection2: ShiftDirection;
39  formulae:
40      GearShifts:
41          (receiveEngineSpeed (engineSpeed) and
              receiveVehicleSpeed (deltaVehicleSpeed,
              vehicleSpeed) and engineSpeed >= 3000 and
              vehicleSpeed > 0 implies gearShiftWidth1 = 1
              and gearShiftDirection1 = Up and
              controlGearShift (gearShiftWidth1,
              gearShiftDirection1)) and
42          (receiveEngineSpeed (engineSpeed) and all
              deltaVehicleSpeed, vehicleSpeed (not
              receiveVehicleSpeed (deltaVehicleSpeed,
              vehicleSpeed)) and engineSpeed >= 3000 and
              storedVehicleSpeed > 0 implies gearShiftWidth1
              = 1 and gearShiftDirection1 = Up and
              controlGearShift (gearShiftWidth1,
              gearShiftDirection1)) and
43          (all engineSpeed (not receiveEngineSpeed
              (engineSpeed)) and receiveVehicleSpeed
              (deltaVehicleSpeed, vehicleSpeed) and
              storedEngineSpeed >= 3000 and vehicleSpeed > 0
              implies gearShiftWidth1 = 1 and
              gearShiftDirection1 = Up and controlGearShift
              (gearShiftWidth1, gearShiftDirection1)) and
44          (receiveEngineSpeed (engineSpeed) and
              receiveVehicleSpeed (deltaVehicleSpeed,
              vehicleSpeed) and engineSpeed <= 1500 and
              deltaVehicleSpeed <= 0 implies gearShiftWidth1
              = 1 and gearShiftDirection1 = Down and
              controlGearShift (gearShiftWidth1,
```

```
            gearShiftDirection1)) and
45          (receiveEngineSpeed (engineSpeed) and all
               deltaVehicleSpeed, vehicleSpeed (not
               receiveVehicleSpeed (deltaVehicleSpeed,
               vehicleSpeed)) and engineSpeed <= 1500 and
               storedDeltaVehicleSpeed <= 0 implies
               gearShiftWidth1 = 1 and gearShiftDirection1 =
               Down and controlGearShift (gearShiftWidth1,
               gearShiftDirection1)) and
46          (all engineSpeed (not receiveEngineSpeed
               (engineSpeed)) and receiveVehicleSpeed
               (deltaVehicleSpeed, vehicleSpeed) and
               storedEngineSpeed <= 1500 and
               (deltaVehicleSpeed <= 0 or vehicleSpeed = 0)
               implies gearShiftWidth1 = 1 and
               gearShiftDirection1 = Down and controlGearShift
               (gearShiftWidth1, gearShiftDirection1)) and
47          (receiveEngineSpeed (engineSpeed) and
               receiveVehicleSpeed (deltaVehicleSpeed,
               vehicleSpeed) and engineSpeed <= 1500 and
               deltaVehicleSpeed > 0 implies all
               gearShiftWidth1, gearShiftDirection1 (not
               controlGearShift (gearShiftWidth1,
               gearShiftDirection1))) and
48          (receiveEngineSpeed (engineSpeed) and all
               deltaVehicleSpeed, vehicleSpeed (not
               receiveVehicleSpeed (deltaVehicleSpeed,
               vehicleSpeed)) and engineSpeed <= 1500 and
               storedDeltaVehicleSpeed >= 0 and
               storedVehicleSpeed > 0 implies all
               gearShiftWidth1, gearShiftDirection1 (not
               controlGearShift (gearShiftWidth1,
               gearShiftDirection1))) and
49          (all engineSpeed (not receiveEngineSpeed
               (engineSpeed)) and receiveVehicleSpeed
               (deltaVehicleSpeed, vehicleSpeed) and
               storedEngineSpeed <= 1500 and deltaVehicleSpeed
               >= 0 and vehicleSpeed > 0 implies all
               gearShiftWidth1, gearShiftDirection1 (not
               controlGearShift (gearShiftWidth1,
               gearShiftDirection1))) and
50          (receiveEngineSpeed (engineSpeed) and
               receiveVehicleSpeed (deltaVehicleSpeed,
```

```
            vehicleSpeed) and engineSpeed >= 1500 and
            engineSpeed < 3000 implies all gearShiftWidth1,
            gearShiftDirection1 (not controlGearShift
            (gearShiftWidth1, gearShiftDirection1))) and
51      (receiveEngineSpeed (engineSpeed) and all
            deltaVehicleSpeed, vehicleSpeed (not
            receiveVehicleSpeed (deltaVehicleSpeed,
            vehicleSpeed)) and engineSpeed >= 1500 and
            engineSpeed < 3000 implies all gearShiftWidth1,
            gearShiftDirection1 (not controlGearShift
            (gearShiftWidth1, gearShiftDirection1))) and
52      (all engineSpeed (not receiveEngineSpeed
            (engineSpeed)) and receiveVehicleSpeed
            (deltaVehicleSpeed, vehicleSpeed) and
            storedEngineSpeed >= 1500 and storedEngineSpeed
            < 3000 implies all gearShiftWidth1,
            gearShiftDirection1 (not controlGearShift
            (gearShiftWidth1, gearShiftDirection1))) and
53      (all engineSpeed (not receiveEngineSpeed
            (engineSpeed)) and all deltaVehicleSpeed,
            vehicleSpeed (not receiveVehicleSpeed
            (deltaVehicleSpeed, vehicleSpeed)) implies all
            gearShiftWidth1, gearShiftDirection1 (not
            controlGearShift (gearShiftWidth1,
            gearShiftDirection1)));
54
55  ReceivingEventAction:
56      all deltaVehicleSpeed1, vehicleSpeed1
            (receiveVehicleSpeed (deltaVehicleSpeed1,
            vehicleSpeed1) implies Until
            (storedDeltaVehicleSpeed = deltaVehicleSpeed1
            and storedVehicleSpeed = vehicleSpeed1, ex
            deltaVehicleSpeed2, vehicleSpeed2
            (receiveVehicleSpeed (deltaVehicleSpeed2,
            vehicleSpeed2)))) and
57      all engineSpeed1 (receiveEngineSpeed
            (engineSpeed1) implies Until (storedEngineSpeed
            = engineSpeed1, ex engineSpeed2
            (receiveEngineSpeed (engineSpeed2))));
58
59  MutualExclusions:
60      all gearShiftWidth1, gearShiftDirection1
            (controlGearShift (gearShiftWidth1,
```

```
            gearShiftDirection1) implies all
            gearShiftWidth2, gearShiftDirection2
            (gearShiftWidth1 <> gearShiftWidth2 and
            gearShiftDirection1 <> gearShiftDirection2
            implies not controlGearShift (gearShiftWidth2,
            gearShiftDirection2))) and
61          all engineSpeed1 (receiveEngineSpeed
            (engineSpeed1) implies all engineSpeed2
            (engineSpeed2 <> engineSpeed1 implies not
            receiveEngineSpeed (engineSpeed2))) and
62          all deltaVehicleSpeed1, vehicleSpeed1
            (receiveVehicleSpeed (deltaVehicleSpeed1,
            vehicleSpeed1) implies all deltaVehicleSpeed2,
            vehicleSpeed2 (deltaVehicleSpeed2 <>
            deltaVehicleSpeed1 and vehicleSpeed2 <>
            vehicleSpeed1 implies not receiveVehicleSpeed
            (deltaVehicleSpeed2, vehicleSpeed2)));
63
64  end
```

6 Annotations

During the last phase of our modelling we decided not to formalize the *Torque Converter* and this decision depends on the way the Torque Converter works.

The Torque Converter is a mechanical component that works coupling and decoupling the *Transmission Shaft* and the *Engine Shaft*. It solves is duty without the necessity to receive commands from any component of the system and this is the cause we have decided to remove it from our model.

Anyway, the state of the Torque Converter is really important for the system since it gives information that permits to insert or not to insert some gears and other details that aren't taken into account in this project.

7 Properties

Listing 8: Property 1

```
actualGear = First and controlGearShift (1, Up) implies
    Futr (actualGear = Second, fluidPropagationDelay +
    singleGearShiftDelay)
```

Listing 9: Property 2

```
gear = Park and gearHandle (gear) and Futr (actualGear =
    Park, fluidPropagationDelay + parkGearShiftDelay)
    implies NowOn (transmissionShaftState = Detached,
    fluidPropagationDelay + parkGearShiftDelay)
```